

# Reading Larger than Usual MCU Tables

D/N : HA0002E

## Introduction

In the MCU application, the look-up table instructions are frequently used. Holtek's 8-bit MCU have two look-up table instructions, namely instructions "TABRDC" and "TABRDL". "TABRDC" is used to look up the table in the current page and "TABRDL" is used to look up the table in the last page. However, the most these two instructions can read is within one page (one page=256 words). If there are more than 256 words, then the table becomes complicated. For example, in the voice processing and LCD display, the look-up table operation is frequently used. More often though, the table content is greater than 256 words, hence, this Application Notes introduce the look-up table program, TABRD, with large capacity table contents, the maximum of which is 32512 (7F00H). Any location in the ROM can be used as a look-up table. However, if the program ROM is more than 8K (for example, HTG21 series, HT48XA3Ö etc.), two table pointers must first be setup by placing the lower-order address of the look-up table to be retrieved in the Table Pointer Register TBLP. Therefore, TABRD is suitable for MCU applications with program ROM<8K.

## Operating Description

We can use the "TABRDC" instruction to complete this function. Since the "TABRDC" instruction is used to read the current page of the table contents, so there must be a "TABRDC" instruction in each page of the table contents. We can use a macro instruction "mENDPAGE" to realize this function.

```
mENDPAGE table, label
```

This table in the macro instruction stands for the beginning address, label stands for the tag of a table jumping destination for table look-up. To complete the large page table, we have to put the macro instruction "mENDPAGE" at the end of each page.

At this stage, we have to carry out the TABRD process. In the TABRD, we can use DataLo and DataHi to specify the table address and the contents read will be stored to the DataLo and DataHi. In this program, the multi "JMP" instruction is used to carry out jump execution to different table locations.  
For detailed information, refer to the Appendix.

## System Resource Options

ROM : 23+ table page  
RAM : 3 ('DataHi', 'DataLo' and 'PageNo')

## Appendix

```

;-----
;File name : largetbl.asm
;Purpose:large type Table Read
;-----
include HT48R30A-1.inc
DATA_LO EQU 1h           ;setup DATA_LO, DATA_HI to verify if
                        ;table location is correct
DATA_HI EQU 2h           ;Table Look-up contents address in 201h

data .section 'data'
DataLo db ?             ;locate table high-byte address,
                        ;return to table high byte contents
DataHi db ?             ;locate table low-byte address,
                        ;return to table low byte contents
PageNo db ?            ;table page number register

mENDPAGE macro table,label ;macro instructions
    org 0feh or (0ff00h and ($-table)) ;Confirm stored table
    ;position
label :
    tabrdc DataLo ;look-up table, return to low-byte
    ;table contents
    jmp _lBackHere ; return to look-up table
endm

code .section at 0 'code' ;main program section
    mov A,DATA_LO ;locate look-up table address
    mov DataLo,A
    mov A,DATA_HI
    mov DataHi,A
    call TABRD ;call look-up table sub-routine
    jmp $

```

```

procedure .section 'code'      ;subroutine area
;-----
;TABRD
;Purpose: read table up to 7f00h (32512) elements
;Input: 'DataHi', 'DataLo' point to table address
;Output: table contents return to 'DataHi', 'DataLo'
;Resources:
;ROM: 23+table page number
;RAM: 3 ('DataHi', 'DataLo' and 'PageNo')
;-----
TABRD proc          ;look-up table subroutine
;Calculate the table page number
;PageNo=DataHi+HIBYTE [DataLo+2*(DataHi+1)]
    mov     A,DataHi
    mov     PageNo,A
    inc     PageNo
    clr     C
    rlc     PageNo
    mov     A,DataLo
    add     A,PageNo
    clr     PageNo
    sz      C
    inc     PageNo
    mov     A,DataHi
    addm    A,PageNo
;Calculate the table location in the current page
    TBLP=DataLo+2*PageNo
    mov     A,PageNo
    mov     TBLP,A
    clr     C
    rlc     TBLP
    mov     A,DataLo
    addm    A,TBLP
;Calculate the correct table address and read the table contents
    mov     A,PageNo
    addm    A,PCL
;
;table jump loop
;
    jmp     _lPage0
    jmp     _lPage1
    jmp     _lPage2
;
_lBackHere:        ;return to look-up table
    mov     A,TBLH    ;read the high-byte table contents
    mov     DataHi,A
    ret
TABRD endp

```

```
-----  
; end of TABRD  
-----  
table .section page 'code' ;table area  
table :  
    dc 000h,001h,002h,.....0fbh,0fch,0fdh ;table of 254 bytes  
    mENDPAGE table,_lPage0  
; fill mENDPAGE to end page data  
    dc 0feh,0ffh,100h,.....1f9h,1fah,1fbh ;table of 254 bytes  
    mENDPAGE table,_lPage1  
    dc 1fch,1fdh,1feh,1ffh,200h,201h  
    mENDPAGE table,_lPage2  
-----  
;End of largetbl.asm  
-----
```